Solve heat equation using Fourier Serie

Solve u_t(t,x) = alpha * u_xx(t,x), u(t,0)=u(t,1)=0, u(0,x)=u_0(x)=x*(1-x)

```
> PDE := diff(u(t,x),t) = alpha * diff(u(t,x),x,x) ;
```

$$PDE := \frac{\partial}{\partial t} u(t,x) = \alpha \left(\frac{\partial^2}{\partial x^2} u(t,x) \right) \quad (1)$$

```
> INI := u0 = x*(1-x)^2 ;
```

$$INI := u0 = x(1-x)^2 \quad (2)$$

Build all the possibile simple solution using separation of variable

```
> U := T(t)*X(x) ;
```

$$U := T(t)\,X(x) \quad (3)$$

Use U in the PDE

```
> subs(u(t,x)=U,PDE) ; EQ1 := expand(%) ;
```

$$\frac{\partial}{\partial t} (T(t)\,X(x)) = \alpha \left(\frac{\partial^2}{\partial x^2} (T(t)\,X(x)) \right)$$

$$EQ1 := \left(\frac{d}{dt} T(t) \right) X(x) = \alpha\, T(t) \left(\frac{d^2}{dx^2} X(x) \right) \quad (4)$$

On the left an ODE depending only on t, on the right an ODE depending only on x

```
> EQ2 := EQ1 / T(t) / X(x) ;
```

$$EQ2 := \frac{\frac{d}{dt} T(t)}{T(t)} = \frac{\alpha \left(\frac{d^2}{dx^2} X(x) \right)}{X(x)} \quad (5)$$

Thus, rhs and lhs are constants, for example C

```
> ODE1 := lhs(EQ2) = C ;
  ODE2 := rhs(EQ2) = C ;
```

$$ODE1 := \frac{\frac{d}{dt} T(t)}{T(t)} = C$$

$$ODE2 := \frac{\alpha \left(\frac{d^2}{dx^2} X(x) \right)}{X(x)} = C \quad (6)$$

Solve the differential equation ODE1 and ODE2

```
> SOL1 := dsolve( ODE1 ) ;
  SOL2 := dsolve( ODE2 ) ;
```

$$SOL1 := T(t) = \_C1\, e^{C t}$$

$$SOL2 := X(x) = \_C1\, e^{\frac{\sqrt{C}\,x}{\sqrt{\alpha}}} + \_C2\, e^{-\frac{\sqrt{C}\,x}{\sqrt{\alpha}}} \quad (7)$$

The general solution (for the ODEs) is:

```
> UC := subs(SOL2,subs(subs(_C1=1,SOL1),T(t)*X(x)));
```

$$UC := e^{C\,t}\left(\_C1\, e^{\frac{\sqrt{C}\,x}{\sqrt{\alpha}}} + \_C2\, e^{-\frac{\sqrt{C}\,x}{\sqrt{\alpha}}}\right) \quad \textbf{(8)}$$

Setup for the boundary conditions

```
> EQBC1  := subs(x=0,UC)=0 ;
  EQBC2  := subs(x=1,UC)=0 ;
```

$$EQBC1 := e^{C\,t}\left(\_C1\, e^{0} + \_C2\, e^{0}\right) = 0$$

$$EQBC2 := e^{C\,t}\left(\_C1\, e^{\frac{\sqrt{C}}{\sqrt{\alpha}}} + \_C2\, e^{-\frac{\sqrt{C}}{\sqrt{\alpha}}}\right) = 0 \quad \textbf{(9)}$$

If C>= 0 the only solution is _C1 = _C2 = 0

```
> solve( {EQBC1,EQBC2}, {_C1, _C2} ) ;
```

$$\{\_C1 = 0, \_C2 = 0\} \quad \textbf{(10)}$$

If C<0 there are peridic solutions

```
> solve( {EQBC1}, {_C2} ) ; EQBC3 := collect(subs( %, EQBC2 ),_C1) ;
```

$$\{\_C2 = -\_C1\}$$

$$EQBC3 := e^{C\,t}\left(e^{\frac{\sqrt{C}}{\sqrt{\alpha}}} - e^{-\frac{\sqrt{C}}{\sqrt{\alpha}}}\right)\_C1 = 0 \quad \textbf{(11)}$$

Extract the part of EQBC3 which must be 0

```
> EQBC3_bis := op(2,lhs(EQBC3));
```

$$EQBC3\_bis := e^{\frac{\sqrt{C}}{\sqrt{\alpha}}} - e^{-\frac{\sqrt{C}}{\sqrt{\alpha}}} \quad \textbf{(12)}$$

Find C such that EQBC3_bis = 0

```
> simplify(subs(C=-omega,EQBC3_bis)) assuming k::integer, k>0,
  omega>0;
```

$$2\,I\,\sin\left(\frac{\sqrt{\omega}}{\sqrt{\alpha}}\right) \quad \textbf{(13)}$$

```
> convert(exp(I*omega),trig);
```

$$\cos(\omega) + I\,\sin(\omega) \quad \textbf{(14)}$$

All the values of omega that satisfy EQBC3_bis are:

```
> COND_ON_omega := sqrt(omega)/sqrt(alpha) = k*Pi ;
```

$$COND\_ON\_omega := \frac{\sqrt{\omega}}{\sqrt{\alpha}} = k\,\pi \quad \textbf{(15)}$$

```
> OMEGA_SOL := solve( COND_ON_omega, {omega} ) ;
```

$$OMEGA\_SOL := \{\omega = k^2\,\pi^2\,\alpha\} \quad \textbf{(16)}$$

Putting all together: C = -omega and omega must satisfy $\frac{\sqrt{\omega}}{\sqrt{\alpha}} = k\,\pi$

```
> UC ;
```

(17)

$$e^{C t}\left(\_C1\, e^{\frac{\sqrt{C}\, x}{\sqrt{\alpha}}} + \_C2\, e^{-\frac{\sqrt{C}\, x}{\sqrt{\alpha}}}\right) \tag{17}$$

Substitute _C1 = I, and _C2 = -_C1

```
> UC1 := subs( _C1=-I/2, _C2 = I/2, UC ) ;
```

$$UC1 := e^{C t}\left(-\frac{1}{2}\, I\, e^{\frac{\sqrt{C}\, x}{\sqrt{\alpha}}} + \frac{1}{2}\, I\, e^{-\frac{\sqrt{C}\, x}{\sqrt{\alpha}}}\right) \tag{18}$$

Substitute C = -omega

```
> UC2 := subs( C=-omega, UC1 ) ;
```

$$UC2 := e^{-\omega t}\left(-\frac{1}{2}\, I\, e^{\frac{\sqrt{-\omega}\, x}{\sqrt{\alpha}}} + \frac{1}{2}\, I\, e^{-\frac{\sqrt{-\omega}\, x}{\sqrt{\alpha}}}\right) \tag{19}$$

Substitute omega that satify the boundary conditions

```
> UC3 := simplify( subs( OMEGA_SOL, UC2 ) ) assuming alpha>0, k>0 ;
```

$$UC3 := e^{-k^2 \pi^2 \alpha t} \sin(k \pi x) \tag{20}$$

Thus, UC3 is a function which satisfy the original PDE with the boundary condition but NOT the INITIAL condition!

```
> subs( u(t,x)=UC3, PDE ) ; expand(%) ;
```

$$\frac{\partial}{\partial t}\left(e^{-k^2 \pi^2 \alpha t} \sin(k \pi x)\right) = \alpha\left(\frac{\partial^2}{\partial x^2}\left(e^{-k^2 \pi^2 \alpha t} \sin(k \pi x)\right)\right)$$

$$-\frac{k^2 \pi^2 \alpha \sin(k \pi x)}{e^{k^2 \pi^2 \alpha t}} = -\frac{k^2 \pi^2 \alpha \sin(k \pi x)}{e^{k^2 \pi^2 \alpha t}} \tag{21}$$

Using linearity of the PDE a generic solution composed by UC3's is of the form

```
> USOL := sum( b[k]*UC3, k=1..infinity ) ;
```

$$USOL := \sum_{k=1}^{\infty} b_k\, e^{-k^2 \pi^2 \alpha t} \sin(k \pi x) \tag{22}$$

```
> subs( u(t,x)=USOL, PDE ) ; expand(%) ;
```

$$\frac{\partial}{\partial t}\left(\sum_{k=1}^{\infty} b_k\, e^{-k^2 \pi^2 \alpha t} \sin(k \pi x)\right) = \alpha\left(\frac{\partial^2}{\partial x^2}\left(\sum_{k=1}^{\infty} b_k\, e^{-k^2 \pi^2 \alpha t} \sin(k \pi x)\right)\right)$$

$$-\pi^2 \alpha\left(\sum_{k=1}^{\infty} \frac{b_k k^2 \sin(k \pi x)}{e^{k^2 \pi^2 \alpha t}}\right) = -\pi^2 \alpha\left(\sum_{k=1}^{\infty} \frac{b_k k^2 \sin(k \pi x)}{e^{k^2 \pi^2 \alpha t}}\right) \tag{23}$$

The unknown a[k] are computed to match the initial condition:

```
> eval(subs(t=0,USOL)) ;
```

$$\sum_{k=1}^{\infty} b_k \sin(k \pi x) \tag{24}$$

Use Fourier serie for the initial condition: u0 is extended in [-1,0] by reflection in x and y then extended by periodicity

```
> u0_ext := piecewise( x<0, x*(1+x)^2, x*(1-x)^2 ) ;
```

$$u0\_ext := \begin{cases} x\,(1+x)^2 & x < 0 \\ x\,(1-x)^2 & otherwise \end{cases} \tag{25}$$

```
> plot( x*(1-x)^2,x=0..1 ) ;
  plot( u0_ext, x=-1..1 ) ;
```





Compute Fourier coeffs for u0_ext

```
> a[0] := int( u0_ext, x=-1..1 ) ;
```

$$a_0 := 0 \quad \text{(26)}$$

```
> a[k] := int( u0_ext*cos(k*Pi*x), x=-1..1 ) ;
```

$$a_k := 0 \quad \text{(27)}$$

```
> b[k] := int( u0_ext*sin(k*Pi*x), x=-1..1 ) ; b[k] := simplify(%)
  assuming k::integer ;
```

$$b_k := \frac{4\left(\cos(k\pi)\,k\pi - 3\sin(k\pi) + 2\,k\pi\right)}{k^4\,\pi^4}$$

$$b_k := \frac{4\left((-1)^k + 2\right)}{k^3\,\pi^3} \quad \text{(28)}$$

```
> USOL ;
```

$$\sum_{k=1}^{\infty} \frac{4\left((-1)^k + 2\right) \mathrm{e}^{-k^2\pi^2\alpha t} \sin(k\pi x)}{k^3\,\pi^3} \quad \text{(29)}$$

```
> U0_check := subs(t=0,USOL) ;
```

$$U0\_check := \sum_{k=1}^{\infty} \frac{4\left(\left(-1\right)^{k}+2\right) \mathrm{e}^{0} \sin\left(k \pi x\right)}{k^{3} \pi^{3}} \tag{30}$$

```
> plot( [U0_check,x*(1-x)^2+0.001], x=0..1 ) ;
```